
第 6 章  常用表单控件的使用 
面向对象程序设计中的对象是由容器类对象和控件类对象组成的，这些对象以可视化控件的形式

组成了表单的具体元件。Visual FoxPro 提供了 20 多种控件供设计时使用，根据这些控件的基本使用功

能大致可分为输出类、输入类、控制类、容器类、连接类 5 大类控件。以上的分类着眼于控件的基本

功能，实际上每个控件兼有多种功能。例如，大部分控件都可以起控制作用，因为它们都含有单击事

件；又如，文本框既能输入也能输出；等等。 
本章将逐一介绍这些控件的基本用法。 

6.1  输出类控件 
输出显示类控件主要有标签（Label）、图像（Image）、线条（Line）和形状（Shape）。这些控件

有各自的属性和事件，虽然标签、图像、线条和形状也定义了诸如单击（Click）、双击（DblClick）等

事件，但一般较少使用。在此重点介绍输出类控件的属性。 

6.1.1  标签 

标签（Label）在控件工具栏中的图标是 。按照标签放入表单

的顺序，标签的默认名称为 Label1、Label2 等。 
标签的主要功能是在表单上显示说明或提示的文本信息，这与

日常的标签类似。标签没有数据源，在表单运行中不能获得焦点，

因此，它显示的内容（即 Caption 的值）在运行期间不能被修改。 
标签的常用属性如表 6-1 所示。需要指出的是这里列出的大部

分属性也是其他控件的常用属性，以后不再重复。 
图 6-1 清楚地说明了 Left、Top、Height、Width 属性。 

表 6-1  标签控件常用属性 

属性名称 功能说明 取值范围 

Caption 指定显示在对象中的标题文本 字符串 

Alignment  指定与控件有关的文本对齐方式 0—左（默认），1—右，2—居中 

AutoSize 确定控件是否根据其内容的长短自动调整大小 .T.或.F.（默认） 

Left 确定控件左边与其父对象左边的距离 数值，单位为像素 

Top  确定控件顶边与其父对象顶边的距离 数值，单位为像素 

Height  确定控件的高度 数值，单位为像素 

Width  确定控件的宽度 数值，单位为像素 

ForeColor 指定对象中显示文本和图形的前景色 RGB(0,0,0)～RGB(255,255,255) 

BackColor 指定对象中显示文本和图形的背景色 RGB(0,0,0)～RGB(255,255,255) 

FontName 指定显示文本时所用的字体名（字型） 字库中所有字体，默认为宋体 

FontSize  指定显示文本时字体的大小（字号） 自然数，默认值为 9 
FontBold 指定显示文本时是否为粗体 .T.或.F.（默认） 
Visible  确定对象是否可见 .T.（默认）或.F. 
WordWrap  确定控件的文本是否随边界自动地反绕折行 .T.或.F.（默认） 

 
图 6-1  Left、Top、Height、 

Width 属性说明 

6.1.2  图像、线条与形状 

1．图像 
图像（Image）在控件工具栏中的图标是 。按照放入表单的顺序，图像控件的默认名称为 Image1、

Image2 等。 
图像的主要功能是用来显示图片的。它能显示多种格式的图像文件，如.bmp、.gif、.jpg、.ico 等类型。

除图像的大小（Height、Width）及在父对象的位置（Left、Top）等通用属性外，图像还有 Picture、Stretch、
BackStyle 等常用属性，说明如表 6-2 所示。 

表 6-2  图像常用属性 

属性名称 功能说明 取值范围 

Picture 指定显示在控件中的图形文件 图形文件的路径和文件名，可用浏览按钮查找 
Stretch 指定如何对图像进行尺寸调整以适应控件大小 0—裁剪（默认）、1—等比填充、2—变比填充 
BackStyle 确定对象的背景色是否透明 数值（0 或 1），默认值为 1（不透明） 

例如，要显示如图 6-2 所示的图像，就需要用图像控件，设置其 Picture
属性为：图像路径\FOX.bmp（提示：FOX.bmp 图像文件可以用 Windows 的
搜索功能找到后再放入用户自己的文件夹，这样再设置 Picture 属性路径就很

明确了），设置 Stretch 属性为 1—等比填充，设置 BackStyle 属性为 0—透明。 
2．线条和形状 
线条（Line）和形状（Shape）都是图形控件。这里一定要区别开图像控件

和图形控件。图像控件操作的对象是“像”，它不是用户现场画的，而是事先保

存好的图像文件；而图形控件操作的对象是“形”，是在设计或运行时画的图形，如线、几何图形等。 
线条在控件工具栏中的图标是 。按照放入表单的顺序，线条控件的默认名称为 Line1、Line2、

Line3 等。线条控件可以在表单任意两点之间画线，用来修饰表单。它的主要属性有 BorderWidth（线

条的宽度）和 BorderStyle（线条的样式）。 
形状在控件工具栏中的图标是 。按照放入表单的顺序，形状控件的默认名称为 Shape1、Shape2

等。形状控件可以在表单中画圆的或方的图形，也是用来修饰表单的。形状的常用属性如表 6-3 所示。 
表 6-3  形状常用属性 

属性名称 功能说明 取值范围 

Curvature 指定 Shape 控件角的曲率 0～99，0 为直角，99 为圆 
Fillcolor 指定封闭图形的填充颜色 RGB(0,0,0)～RGB(255,255,255) 
FillStyle 指定表单、形状等的填充类型 0～7：0—实线、1—透明（默认）等 
SpecialEffect 指定控件的样式 1—平面（默认），0—三维 
BackStyle 确定对象的背景色是否透明 0—透明，1—不透明（默认） 

例如，要在表单中画一个红色的球，就需要用形状控件，

设置其 Width 和 Height 均为 100，Curvature 为 99，FillStyle
为 0—实线，Fillcolor 为 RGB(255,0,0)（或设置 BackStyle 为 1
—不透明，BackColor 为 RGB(255,0,0)）。 

【例 6.1】设计如图 6-3 所示的一个显示表单。要求：放一

幅图片并配上诗。图片用两个形状作外框，诗句采用竖排、华

文行楷、20 号、白底、蓝字显示。 

设计步骤如下： 
 

图 6-3  例 6.1 的设计界面 

 
图 6-2  显示图像 

① 新建表单，设置表单的 Caption 属性为“诗画”； 
② 放入图像控件，在属性窗口设置其 Picture 属性指向图片，调整图像大小并设置 Stretch 为等比

填充； 
③ 为图像框一个形状，设置形状的 SpecialEffect 为三维，用布局工具栏或格式菜单使形状置后，

调整大小使其正好框住图像，同样再放入另一个形状，本例用了两个形状来修饰图像； 
④ 放入一个标签，设置其 Caption 为“诗句”，设置其 AutoSize 为.T.，WordWrap 为.T.，往左挤

压文字使之竖排，放入其他 3 个标签，同样做上面的设置； 
⑤ 按住“Shift”键选中所有写上诗句的标签，设置它们以华文行楷、20 号、白底、蓝字显示，

然后调整好布局； 
⑥ 保存、运行即可。 

6.2  输入类控件 
输入类控件主要有文本框（TextBox）、编辑框（EditBox）、列表框（ListBox）、组合框（ComboBox）、

微调（Spinner）等。如果将文本框（TextBox）、编辑框（EditBox）的只读属性 ReadOnly 设置为.T.，
通常也可以作为输出显示的控件。 

6.2.1  文本框 

文本框（TextBox）在控件工具栏中的图标是 。按照文本框放入的顺序，文本框的默认名称为

Text1、Text2 等。 
文本框是表单设计中用到较多的控件，它既能接收用户的输入，也能输出指定的信息。文本框可

以设置数据源，允许用户用文本框编辑保存在表中的非 M、非 G 型数据，也可以通过 Value 属性赋值

或读取文本框的当前值。 
1．文本框的常用属性 
文本框的常用属性如表 6-4 所示。 

表 6-4  文本框的常用属性 

属性名称 功能说明 取值范围 

Value 指定控件的当前值 非 M、非 G 型数据 
ControlSource 指定控件的数据源  表的字段、内存变量 
DateFormat 指定控件的日期格式 0～14，0—默认，14—汉语等 
InputMask 指定键入到文本框中字符的特性 设置值 
Format 指定在文本框中值的显示方式 设置值 
ReadOnly 指定控件是否只读 .T.，.F.（默认） 
SelectOnEntry 当控件得到焦点后是否选中该单元 .T.，.F.（默认） 
PassWordChar 用指定的字符替代输入的数据 常用“*” 

InputMask 属性决定了键入到文本框中字符的特性。例如，将 InputMask 属性设置为“999,999.99”
可限制用户只能输入具有两位小数并小于 1 000 000 的数值，在用户输入任何值之前，逗号和点号就显

示在文本框中，如果用户输入一个字母，则这个字母不被接受。表 6-5 列出了常用的 InputMask 属性

的设置值。 
如果有逻辑字段，并且想让用户能键入“.Y.”或“.N.”而不是“.T.”或“.F.”，应将 InputMask

属性设置为“.Y.”。 

表 6-5  InputMask 属性的设置值 

设 置 值 说    明 

X 可以输入任何字符 
9 可以输入数字和正负号 
# 可以输入数字、空格和正负号 
$ 在固定位置显示当前的货币符号（用 SET CURRENCY 指定） 
$$ 显示浮点货币符号 
* 在数字的左边显示星号 
. 指示小数点的位置 

 ， 分隔小数点左边的数字串 

2．文本框的常用方法 
文本框最常用的方法是给文本框设置焦点。当一个对象被选定时称为该对象获得了焦点。文本框获得

焦点的标志是文本框内出现了闪烁的光标，而命令按钮获得焦点的标志则是命令按钮上出现虚线框。 
给控件设置焦点的方法主要有以下 3 种：一是使用 Tab 键，将焦点移到指定的控件上；二是使用

鼠标单击控件；三是可以使用系统的方法程序 SetFocus。 
SetFocus 方法的使用如下。 
格式： 

控件. SetFocus 

功能：为一个控件指定焦点，即确定当前操作的对象。 
应用于：复选框、列、组合框、命令按钮、容器对象、控件对象、编辑框、表格、列表框、OLE

绑定型控件、OLE 容器控件、选项按钮、微调和文本框。 
说明：如果控件的Enabled 或Visible 属性设置为“假”（.F.），或者控件的When 事件返回“假”（.F.），

则不能给一个控件指定焦点，因此在控件使用SetFocus方法之前，必须首先把上面提到的属性值设置为“真”

（.T.）。一旦控件获得了焦点，用户的任何输入都针对这个控件。 
例如： 

Thisform.Text1.SetFocus  &&表单中的 Text1获得焦点 

3．文本框的常用事件 
文本框的常用事件有：Init、GotFocus、LostFocus、Valid、When 等。对于事件先要掌握事件的触

发时机，再根据任务的需要选择事件、写事件代码。事件的触发时机是系统规定好的，但是每个程序

的事件代码是不一样的，不同的任务需要写不同的代码。下面介绍几个常用事件的触发时机。 
① Init：创建对象时发生。 
说明：容器中各对象的 Init 事件在容器的 Init 事件之前触发，所以，对于各对象的初始化处理一

般都集中写在表单的 Init 事件中。 
② When：在控件接收焦点之前此事件发生。 
③ GotFocus：当控件接收到焦点时触发。 
④ Valid：在控件失去焦点之前发生。 
说明：Valid 事件有返回值，默认返回.T.。若 Valid 事件返回.T.或 1，则控件可以失去焦点，继续

后续操作；若返回.F.或 0，则控件不能失去焦点，把光标锁定在控件上，不能进行后续的操作。此事

件常用于输入时检验输入数据的正确性。 
⑤ LostFocus：当控件失去焦点时触发。 
从上面各事件的触发时机可以看出，上述事件发生的先后顺序是：Init、When、GotFocus、Valid、

LostFocus。 

【例 6.2】编写一个口令验证窗口，当输入的口令正确则显示欢迎的提示信息，不正确则要求重新

输入口令。运行界面如图 6-4 所示。 
（1）问题分析 
检验口令的正确性需要编写文本框的 Valid 事件代码。输入的口令用“*”覆盖，设置文本框的

PassWordChar 属性为“*”，为了方便下次输入，设置 SelectOnEntry 为.T.，即选中文本。提示信息用

MESSAGEBOX( )函数实现。 

       
(a)                           (b)                            (c)                          (d) 

图 6-4  口令验证界面 

（2）设计步骤 
① 建立表单，放入标签和文本框控件。 
② 在属性窗口设置属性：表单 Form1 的 Caption、标签的 Caption、文本框 Text1 的 PassWordChar

和 SelectOnEntry 属性。 

③ 编写文本框 Text1 的 Valid 事件代码： 
IF ALLTRIM(This.Value)=="ABC"   &&精确比较,口令为"ABC" 

    =MESSAGEBOX("欢迎使用本系统")   &&利用函数显示提示 
    RETU .T.         &&可以失去焦点，继续后面的操作 

ELSE 

    =MESSAGEBOX("口令错，请再试一次!") 
    RETURN .F.         &&不能失去焦点，光标锁定在文本框上 

ENDIF 

利用文本框实现输出的实例，在第 5 章的例 5.1 中已有应用。 

6.2.2  编辑框 

编辑框（EditBox）在控件工具栏中的图标是 。按照编辑框放入的顺序，编辑框的默认名称为

Edit1、Edit2 等。 
编辑框主要是用来处理长的字符型数据和备注型字段的，它是既能做输入也能做输出的控件。编

辑框与文本框主要区别如下： 
（1）编辑框只能处理字符型数据（备注型实际是长的字符型数据），而文本框可以处理字符、数

值、日期、逻辑等类型的数据； 
（2）编辑框可以处理多段文本，回车不退出，而文本框按回车即退出。 
编辑框的常用属性如表 6-6 所示。 

表 6-6  编辑框的常用属性 

属性名称 功能说明 取值范围 

ControlSource 指定控件的数据源  C 型字段、内存变量或 M 型字段 

Value 指定控件的当前值 字符型数据 

Text 控件中输入的无格式文本 设计时不可用，运行时可读取 

SelText 返回在编辑框中选定的文本 设计时不可用，运行时可读取 

ReadOnly 指定控件是否只读 .T.，.F.（默认） 

例如，要显示和编辑职工表的简历字段，则设置编辑框的 ControlSource 属性为“职工.简历”即

可。清除在 Edit1 中选定的文本可使用命令：Thisform.Edit1. SelText= ""。 
【例 6.3】设计一个表单，能在编辑框中输出一个对角线为 0 其余

为 1 的一个矩阵。运行界面如图 6-5 所示。 
（1）问题分析 
这是一个输出 N 行 N 列二维图形的问题，根据第 4 章学习的知识，

要用双重循环来解此题，即外循环 I 控制输出的行数，内循环 J 控制每

行输出的列数。当 I=J 或 I+J=N+1 时，是对角线上的元素。由于是通

过编辑框输出，因此，先将要输出的内容以字符串的形式保存在一个内存变量中，用函数 CHR(13)实
现回车换行，再把这个长字符串赋值给编辑框。上述处理写在“显示”按钮的 Click 事件中。另外，

控制文本框输入的数字为 1～9，则需要写文本框的 Valid 事件代码。 
（2）设计步骤 
① 新建表单，放入 1 个标签、1 个文本框、1 个命令按钮、1 个编辑框。 
② 在属性窗口设置 Form1、Label1、Command1 的 Caption 属性。 
③ 编写表单 Form1 的 Init 事件。 

Thisform.Command1.Default=.T.   &&设置 Command1为默认按钮 
Thisform.Text1.SelectOnEntry =.T. &&焦点打到文本框时，自动选中文本 
Thisform.Text1.Value=1  &&将文本框设置为数值型数据（文本框默认为字符型数据） 
Thisform.Edit1.ReadOnly =.T.    &&设置编辑框为只读 

④ 编写命令按钮 Command1（显示）的 Click 事件代码。 
N=Thisform.Text1.Value    &&获取文本框的输入值 
P=""                  &&长字符串的初值 
 FOR I=1  TO N      &&外循环控制输出的行数 
   FOR J=1 TO N     &&内循环控制输出的列数 
     IF I=J OR J=N-I+1 &&左对角线 I=J,右对角线 J=N-I+1 
         A=0       &&对角线上元素赋值 0,为输出清晰在零的前面加一空格 
      ELSE 
         A=1      &&非对角线上元素赋值 1 
      ENDIF 
      P=P+STR(A,2)  &&把每个数字转换成字符后连接起来 
   ENDFOR 
   P=P+CHR(10)     &&一行结束加换行符，或回车符 CHR(13) 
 ENDFOR 
Thisform.Edit1.Value=P    &&把带换行符的长字符串赋值给编辑框 
Thisform.Text1.SetFocus   &&焦点打到文本框上，为下次输入作准备 
Thisform.Refresh  

⑤ 编写文本框 Text1 的 Valid 事件代码。 
IF This.Value>9  OR This.Value<1 
  =MESSAGEBOX("输入错,请重新输入!") 
  Return .F. 
ELSE  
  Return .T. 
ENDIF 
Thisform.Refresh 

6.2.3  列表框与组合框 

列表框（ListBox）与组合框（ComboBox）主要用于给用户提供预先设定的多种选项，极大地方

便了用户的输入。例如，学校新生在注册时需要输入省份，如果输入界面提供了省份的列表框，则用

户只需要在其中选择一项即可。 

图 6-5  例 6.3 的运行界面

1．列表框 
列表框在控件工具栏中的图标是 。按照列表框放入的顺序，列表框的默认名称为 List1、List2

等。列表框的功能主要是给用户提供预先设定的多种选项。 
（1）列表框的常用属性 
列表框的常用属性如表 6-7 所示。 

表 6-7  列表框的常用属性 

属性名称 功能说明 取值范围 

RowSourceType 确定控件中的数据源的类型 数值（0，1，2，…，9） 
RowSource 确定列表框的数据源 与 RowSourceType 配合使用 
ListCount 列表框中数据的项数 设计时不可用，运行时可读取 
List(i) 访问列表框中数据项目的数组 i:1～ListCount 
ListIndex 指定列表框中选定数据项的索引号 设计时不可用，运行时可读取 
ColumnCount 列表框的列数 数值 
Value 当前值（必须是列表中存在的） 设计和运行时可用，列表框和组合框是只读的 
DisplayValue 当前值（可以是用户输入的） 设计和运行时可用 
Sorted 指定列表框各项是否按字母顺序排列 .T.，.F.（默认） 

对于属性 RowSourceType 的 10 种选项的说明见表 6-8。 

表 6-8  RowSourceType 属性选项说明 

属 性 值 列表项的数据源 

0—无 默认值。用 AddItem 方法添加列表项，可用 RemoveItem 方法从列表中移去列表项 

1—值 
用 RowSource 属性指定多个要在列表中显示的值。例如：

Thisform.List1.RowSource ="直销,零售,客服" 

2—别名 
在列表中包含打开表的一个或多个字段的值。表由 RowSource 属性指定。

如果 ColumnCount 属性设置为 0 或 1，列表将显示表中第一个字段的值； 
如果 ColumnCount 属性设置为 3，列表将显示表中最前面的 3 个字段值 

3—SQL 语句 
SELECT-SQL 语句的结果作为列表框的数据源。例如：

Thisform.List1.RowSource ="SELECT 姓名 FROM 职工 INTO CURSOR mylist" 

4—查询(.QPR) 
用查询的结果填充列表框，查询是在“查询设计器”中设计的。例如：

Thisform.List1.RowSource = "职工查询.qpr" 

5—数组 
用数组中的项填充列表。例如，如果有一个名为 AAA 的表单数组属性，Thisform.List1.RowSource = 

"Thisform.AAA"  

6—字段 
可以指定一个字段或用逗号分隔的一系列字段值来填充列表。例如：

Thisform.List1.RowSource = "职工.姓名" 

7—文件 
用当前目录下的文件来填充列表。例如，

要在列表中显示 Visual FoxPro 表，可将 RowSource 属性设置为*.dbf 

8—结构 用 RowSource 属性指定的表的结构中的字段名来填充列表 

9—弹出式菜单 用一个先前定义的弹出式菜单来填充列表 

注意，如果 RowSourceType 设置为类型 2（别名）或类型 6（字段），当用户在列表中选择新值时，

表的记录指针将同步移动到相应记录，并且列表框中的项目不能移出。 
（2）列表框的常用方法 
列表框的常用方法有 AddItem、RemoveItem、Clear，它们的使用介绍如下。 
① AddItem 方法 
格式： 

控件.AddItem(cItem [, nIndex] [, nColumn]) 

功能：在组合框或列表框中添加一个新数据项。 
应用于：组合框和列表框。 

说明： 
cItem 指定添加到控件中的字符串表达式。 
nIndex 指定控件中放置数据项的位置。如果忽略参数 nIndex，并且 Sorted 属性设置为.T.，则 cItem
数据按字母排序方式添加到队列；如果忽略参数 nIndex，并且 Sorted 属性设置为“假”（.F.），
则 cItem 将添加到组合框或列表框控件的列表区末尾。 

nColumn 指定控件的列，新数据项加入到此列中，默认值为 1。 
例如，要把职工表（假设职工表已经放入数据环境）中所有的部门加入到表单下的列表框中，不

要重复的部门项出现，代码如下： 
INDEX ON 部门 TO BM UNIQUE 
SCAN  
 Thisform.List1.ADDITEM(部门) 
ENDSCAN 
CLOSE INDEX 

② RemoveItem 方法 
格式： 

控件. RemoveItem ( nIndex) 

功能：从组合框或列表框中移去一项。 
应用于：组合框和列表框。 
说明：nIndex 指定一个整数，即数据项的位置，可以用 ListIndex 属性来指定。 
例如，如果要将表单下的列表框中的第 1 项数据移除，命令为： 

Thisform.List1.RemoveItem(1) 

如果要将表单下的列表框中选定的某一项数据移除，命令为： 
Thisform.List1.RemoveItem(Thisform.List1.ListIndex) 

③ Clear 方法 
格式： 

对象.Clear 

功能：清除列表框或组合框中的文本。 
应用于：列表框和组合框。 

说明：只有当列表框或组合框的数据源类型RowSourceType属性设置为0时，该方法才生效。 
例如，将当前表单中的列表框中的内容清除： 

Thisform.List1.Clear 

（3）列表框的常用事件 
列表框常用到 InterActiveChange 事件，它的触发时机是：在使用键盘或鼠标更改控件的值时，此

事件发生。该事件可应用于：复选框、组合框、命令组、编辑框、列表框、选项组、微调和文本框等。 
【例 6.4】按姓名查询职工的信息。设计界面如图 6-6 所示，要求初始运行时只显示左边的姓

名列表框信息，选择某一姓名后才显示指定姓名的具体信息，如图 6-7 所示。 
（1）问题分析 
表单用到职工表，把职工表放入数据环境。选择姓名用的是列表框，需要设置列表框的数据源，设置

列表框的RowSourceType 为 6，RowSource 为职工.姓名，用这种设置，当用户在列表中选择新值时，表的

记录指针将同步移动到相应记录。右边显示的数据直接从数据环境拖入即可，表中的字段能自动与对应的

控件建立绑定。需要写表单的 Init 事件代码，设置右边的数据不可见。需要写列表框的 InterActiveChange
事件代码，当用户在列表中选择新值后，使右边的数据可见，同时表单以最新的数据显示。 

 图 6-6  例 6.4 设计界面        图 6-7  例 6.4 选择姓名后的界面 

（2）设计步骤 

① 新建表单，把职工表放入数据环境。 

② 放入标签和列表框控件，在属性窗口设置表单和标签的 Caption 属性，设置列表框的

RowSourceType 为 6，RowSource 为职工.姓名。 

③ 把职工号、姓名、性别、婚否、出生日期、部门、基本工资、简历字段从数据环境拖入表单，

逻辑型字段“婚否”自动与复选框绑定，备注型字段简历自动与编辑框绑定，其余字段自动与文本框

绑定。 

④ 编写表单 Form1 的 Init 事件代码。 
Thisform.Setall("Visible",.F.) &&设置表单中所有对象不可见 
Thisform.Label1.Visible=.T.  &&设置 Label1可见 

Thisform.List1.Visible=.T.  &&设置 List1可见 

Thisform.SetAll("ReadOnly",.T.,"TextBox")  &&设置所有文本框为只读 
Thisform.Edt简历.ReadOnly=.T. &&编辑框为只读 

Thisform.Chk婚否.ReadOnly=.T. &&复选框为只读 

⑤ 编写列表框 List1 的 InterActiveChange 事件代码。 
Thisform.Setall("Visible",.T.) &&设置表单中所有对象可见 

Thisform.Refresh    &&刷新表单 

【例 6.5】设计数据在列表框之间转移操作的表单。要求：能把左边列表框选定的数据项移到右边

列表框，也能把右边列表框选定的数据项移到左边列表框。初始运行界面和操作后的界面分别如图 6-8
和图 6-9 所示。 

               
图 6-8  初始运行界面                         图 6-9  操作后的界面 

（1）问题分析 
初始运行时只有左边列表框有数据，所以左边列表框要先设置数据源。数据源填入的是职工的姓

名字段，设置 RowSourceType 为 6，RowSource 为“职工.姓名”，行吗？显然不行，因为左边列表框

的数据要能移出，用RowSourceType为 6是不能移出的，见表 6-7的功能说明。必须设置RowSourceType
为 0，用 AddItem 方法填入数据，这样列表框的数据可以用 RemoveItem 方法移出，也可用 Clear 方法

清除。该表单需要写表单的 Init 事件和两个命令按钮的 Click 事件代码。 

（2）设计步骤 
① 新建表单，将职工表放入数据环境。 
② 用表单控件工具栏，放入 2 个标签、2 个列表框、2 个命令按钮。在属性窗口设置表单和 2 个

命令按钮的 Caption 属性。（左边的列表框为 List1，右边的是 List2。） 

③ 编写表单 Form1 的 Init 事件代码。 
Thisform.List1. RowSourceType=0 

SCAN  &&用表循环把所有姓名填入列表框 

    Thisform.List1.AddItem(职工.姓名) 
ENDSCAN 

④ 编写“选择一项”命令按钮 Command 1 的 Click 事件代码。 
Thisform.List2.AddItem(Thisform.List1.Value)  &&把 List1当前值填入 List2 
Thisform.List1.RemoveItem(Thisform.List1.ListIndex)&&把List1选中的项移出 

Thisform.Refresh 

⑤ 编写“退还一项”命令按钮 Command2 的 Click 事件代码。 
Thisform.List1.AddItem(Thisform.List2.Value) &&把 List2当前值填入 List1 

Thisform.List2.RemoveItem(Thisform.List2.listIndex)&&把List2选中的项移出 

Thisform.Refresh 

在本例操作中，一定要先在 List1 中选中某项数据，即确定 ListIndex 属性后再按“选择一项”命

令按钮，否则表单会出错。如果要解决此问题，可以先设置命令按钮 1 的 Enabled 属性为.F.。只有当

在 List1 中选中一内容，即在 List1 的 Click 事件中，使命令按钮的 Enabled 置.T.。操作 Command2 同

样有上述问题。 
2．组合框 
组合框在控件工具栏中的图标是 。按照组合框放入的顺序，组合框的默认名称为 Combo1、

Combo2 等。组合框既能输入数据也能选择数据，兼有文本框和列表框的功能，可以把组合框看成是文

本框与列表框的组合。但是请注意，如果把组合框的 Style 属性（用于指定控件的样式）设置为“2—下

拉列表框”时，这时的组合框只有列表框的功能，即只能当作列表框来使用。组合框 Style 属性的默认

值是“1—下拉组合框”，也就是说组合框的 Style 属性为“1”时才具有文本框和列表框的功能。组合框

的常用属性除上面提到的 Style 属性外，还有RowSourceType、RowSource、Value、DisplayValue、Sorted、
ControlSource、PassWordChar 和 SelectOnEntry 等，可以看出这些是列表框与文本框的常用属性，它们的

说明见表 6-4 和表 6-7。 
与列表框一样，组合框常用的方法也是 AddItem、RemoveItem、Clear 等，它们的使用方法与前面

一样，不再赘述。 
组合框常用的事件有 Valid、InterActiveChange 等。 
【例 6.6】根据“职工”表，设计一个计算指定部门基本工资平均数的表单。要求：指定部门可以

由用户输入也可以从列表中选择，如果用户输入的部门不存在，则显示“查无此部门！”，如果存在则

计算指定部门基本工资的平均数，并显示在文本框中。设计界面如图 6-10 所示，运行界面如图 6-11、
图 6-12 和图 6-13 所示。 

               
   图 6-10  例 6.6 设计界面                        图 6-11  例 6.6 初始运行界面 

       图 6-12  例 6.6 选择部门后的界面             图 6-13  例 6.6 输入部门不存在时的界面 

（1）问题分析 
对于部门数据的输入，要求既能由用户自己输入，也可以从列表中选择输入，所以本例采用组合

框控件。部门数据的列表项不是简单地将部门字段绑定，而是将重复的部门去掉后，循环运用 AddItem
方法来填入部门的。表单初始运行时 Label2、Text1 不显示。另外，执行统计计算命令时需要关闭人

机对话功能，不然表单上会出现干扰信息等，这些处理都应在表单的 Init 事件里完成。利用组合框的

Valid 事件可以有效地检验输入是否正确。计算指定部门基本工资平均数的处理用组合框的

InterActiveChange 事件比较恰当。 
（2）设计步骤 
① 新建表单，将职工表放入数据环境。 
② 放入 2 个标签、1 个组合框、1 个文本框和 1 个命令按钮。在“属性”窗口设置表单和 Label1

的 Caption 属性，设置组合框的 SelectOnEntry 为.T.，设置命令按钮的 Caption。 

③ 表单 Form1 的 Init 事件代码如下： 
SET TALK OFF        &&关闭系统应答 
SET SAFETY OFF        &&关闭安全开关 
Thisform.Combo1.RowSourceType=0    &&设置组合框的 RowSourceType为 0 
INDEX ON 部门 TO BM UNIQUE      &&按部门建立唯一索引 
SCAN   &&用表循环把所有部门填入到组合框 
  Thisform.Combo1.AddItem(部门) 
ENDSCAN 
CLOSE INDEX        &&关闭部门唯一索引 
Thisform.Label2.Visible=.F.    &&设置 Label2不可见 
Thisform.Text1.Visible=.F.     &&设置 Text1不可见 

④ 组合框 Combo1 的 Valid 事件代码如下： 
LOCATE FOR ALLTRIM(Thisform.Combo1.DisplayValue)=ALLTRIM(部门) &&查找部门 
IF EOF()       &&到文件尾，则不存在 
   Thisform.Label2.Visible=.F.    &&设置 Label2不可见 
   Thisform.Text1.Visible=.F.     &&设置 Text1不可见 
   =MESSAGEBOX("查无此部门！")     &&显示提示信息 
   RETURN .F. 
ELSE   &&否则找到部门 
   Thisform.Label2.Visible=.T.    &&设置 Label2可见 
   Thisform.Text1.Visible=.T.     &&设置 Text1可见 
   RETURN .T. 
ENDIF 
Thisform.Refresh 

⑤ 组合框 Combo1 的 InterActiveChange 事件代码如下： 
**计算指定部门工资的平均数 

AVERAGE 基本工资 TO GZ FOR ALLTRIM(Thisform.Combo1.DisplayValue)= ALLTRIM (部门)  

**设置 Label2的标题 
Thisform.Label2.Caption=ALLTRIM(Thisform.Combo1.DisplayValue)+"部基本工资平均数"   
Thisform.Text1.Value=GZ   &&填写结果 

Thisform.Refresh 

⑥ 命令按钮 Command1（退出）的 Click 事件代码如下： 
SET TALK ON       &&打开系统应答 
SET SAFETY ON       &&打开安全开关 
Thisform.Release 

6.2.4  微调 

微调（Spinner）控件在控件工具栏中的图标是 。按照微调控件放入的顺序，微调控件的默认名

称为 Spinner1、Spinner2 等。微调控件的功能主要是用来选择或输入一定范围的数值型数据，它既可

以由键盘输入数据，也允许通过微调的向上或向下箭头对微调控件中的当前值进行增减操作。 
1．微调的常用属性 
微调的常用属性如表 6-9 所示。 

表 6-9  微调常用属性 

属性名称 功能说明 取值说明 

Increment 用户每次单击向上或向下按钮时，微调框增加和减

少的数值 
数值，默认值为1；设置为正时，单击向上/向下按

钮，Value对应增加/减少，设置为负时反之 
KeyboardHighValue 用户通过键盘能输入到微调框中的最高值 数值，应大于 KeyboardLowValue 
KeyboardLowValue 用户通过键盘输入到微调框中的最低值 数值，应小于 KeyboardHighValue 
SpinnerHighValue 用户单击向上按钮时，微调控件能显示的最高值 数值 
SpinnerLowValue 用户单击向下按钮时，微调控件能显示的最低值 数值 
value 微调文本框的当前值 数值 

2．微调的常用事件和方法 
微调常用的事件与文本框类似，主要有：Init、When、GotFocus、Click、Valid、LostFocus、

InterActiveChange 和 KeyPress 等，常用方法有 Refresh 和 SetFocus。此外，微调还有如下两个最为常

用的事件。 
DownClick 事件：单击向下箭头按钮时触发。 
UpClick 事件：单击向上箭头按钮时触发。 
【例 6.7】编制一手工日历。设计界面与运行界面如图 6-14 和图 6-15 所示。 

             
                  图 6-14  手工日历设计界面              图 6-15  手工日历运行界面 

（1）问题分析 
微调框不能用来直接显示日期型数据，但将微调与文本框结合起来就可以显示多种类型的数据了。

用文本框来显示日期型数据，通过微调的DownClick、UpClick 事件实现上下翻动。 
（2）设计步骤 
① 设置表单的标题属性 Caption，放入 1 个文本框来显示日期，放入一个微调，利用其 DownClick

和 UpClick 事件。 
② 表单 Form1 的 Init 事件代码如下： 

Thisform.Text1.Value=Date() 
Thisform.Text1.DateFormat=14 &&设置文本框的日期格式为汉语 
Thisform.Spinner1.SetFocus  &&焦点移到微调，使文本框的日期格式激活 

③ 微调 Spinner1 的 DownClick 事件代码如下： 
Thisform.Text1.Value=Thisform.Text1.Value-1 
Thisform.Refresh 

④ 微调 Spinner1 的 UpClick 事件代码如下： 
Thisform.Text1.Value=Thisform.Text1.Value+1 
Thisform.Refresh 

6.3  控制类控件 
用户通过控制类控件可以触发表单中的某一事件代码，从而控制程序执行指定的动作。控制类控件主

要有命令按钮、命令按钮组、选项按钮组和计时器控件等。下面分别介绍这些控件。 

6.3.1  命令按钮与命令按钮组 

1．命令按钮 
命令按钮（CommandButton）在控件工具栏中的图标是 。按照命令按钮放入表单的顺序，命

令按钮的默认名称为 Command1、Command2 等。 
命令按钮是可视化程序设计中最常用的一种控件。它的功能主要是通过用户单击命令按钮来完成

特定的任务，对于不同的任务通过不同的命令按钮来实现，程序设计者需要做的就是给命令按钮恰当

的标题，并且为完成这些任务编写事件代码。 
命令按钮的常用属性除 Width、Height、FontName、FontSize、ForeColor 等基本属性外，还有一

些常用属性，如表 6-10 所示。 

表 6-10  命令按钮的常用属性 

属性名称 功能说明 取值范围 

Caption 指定显示在对象中的标题文本 字符串。可在标题字符串中加 “\<字母”的方法设置快捷键 
Default 指定命令按钮是否为默认按钮 .T.（按 Enter 键触发此按钮）或.F.（默认） 
Enabled 指定命令按钮是否响应用户触发的事件 .T.（默认）或.F.（按钮为灰色） 
Picture 指定命令按钮上显示的图形文件 .ico、.bmp、.gif、.jpg 等 
ToolTipText 指定命令按钮的提示文本 字符串 
WordWrap  指定按钮标题文本是否折行显示 .T.或.F.（默认） 

当有多个按钮时，一般把最常用的那个按钮设置为默认按钮，按回车键时默认按钮即被触发。默

认按钮与其他按钮的区别是有阴影边框，如图 6-16 所示，把按钮的 Default 属
性设置为.T.即可。我们在日常使用计算机时经常碰到默认按钮。 

使用命令按钮时常用到的方法程序是 SetFocus，即将焦点打到指定的命令

按钮上。例如： 
Thisform.Command1.SetFocus  &&把光标打在表单下的 Command1上 

就像平时使用命令按钮基本上是采用单击的动作一样，命令按钮的常用事件主要是Click（单击），该

事件在单击命令按钮时触发。可以说命令按钮一般都需要编写其Click 事件代码。 
【例6.8】设计一个职工表的只读浏览表单，用户能通过前翻、后翻命令按钮翻页记录，当翻到文件头

时，前翻按钮禁止使用，翻到文件尾时，后翻按钮禁止使用。运行

界面如图 6-17 所示。 
（1）问题分析 
题目要求显示职工表数据，所以控件需要与字段绑定，最简

单的方法是用数据环境，把字段直接拖入表单。把控件的

ReadOnly 设置为.T.即实现只读。根据要求需要编写 Command1、
Command2 的 Click 事件代码。 

 
图 6-16  默认按钮 

 
图 6-17  例 6.8 运行界面 

（2）设计步骤 
① 新建表单，把职工表放入数据环境，然后把职工号、姓名、性别等字段拖入表单，它们会与

对应的控件建立绑定。然后放入 3 个命令按钮，调整布局及字体大小。 
② 在“属性”窗口设置以下属性：表单及命令按钮的 Caption；把 Command2 的 Default 属性设

置为.T.，使其为默认按钮；按住 Shift 键，选中所有的文本框、复选框、编辑框，设置它们的 ReadOnly
为.T.。 

③ 编写命令按钮 Command1（前翻）的 Click 事件代码如下： 
SKIP –1  &&下移一条记录 
IF BOF() &&如果是文件头 
  Thisform.Command1.Enabled=.F.   &&Command1禁止使用 
ENDIF 
Thisform.Command2.Enabled=.T.    &&设置 Command2能使用 
Thisform.Refresh 

④ 编写命令按钮 Command2（后翻）的 Click 事件代码如下： 
SKIP  &&上移一条记录 
IF EOF() &&如果是文件尾 
   Thisform.Command2.Enabled=.F.   &&Command2禁止使用 
ENDIF 
Thisform.Command1.Enabled=.T.    &&设置 Command1能使用 
Thisform.Refresh 

⑤ 编写命令按钮 Command3（退出）的 Click 事件代码如下： 
Thisform.Release 

2．命令按钮组 
命令按钮组（CommandGroup）在控件工具栏中的图标是 。按照命令按钮组放入表单的顺序，

命令按钮组的默认名称为 CommandGroup1、CommandGroup2 等。 
命令按钮组的功能也是通过用户单击命令按钮来完成特定的任务。只不过由于命令按钮组是容器

类对象，其中可以包含有多个命令按钮，因此，如果一个表单里要用到多个命令按钮，则可以用一个

命令按钮组来实现。 
命令按钮组的常用属性如表 6-11 所示，由于命令按钮组属于容器这个大类，它的记数属性

（ButtonCount）和收集属性（Buttons(i)）是经常用到的属性。 

表 6-11  命令按钮组的常用属性 

属性名称 功能说明 取值范围 

ButtonCount 指明按钮组中命令按钮的数目 正整数，默认为 2 
Buttons(i) 用于引用每个按钮的数组  i:1～ButtonCount 
Value 指明用户按了哪个按钮  正整数：1～ButtonCount 
BackStyle 命令按钮组的背景风格 0—透明，1—不透明（默认） 

在使用命令按钮组时，可以利用 Visual FoxPro 提供的命令按钮组生成器，方便地设置按钮的个

数、按钮的标题及按钮的布局等。可以用命令按钮组的快捷菜单进入其生成器，生成器的对话框界

面如图 6-18 所示。   

使用命令按钮组时常用到的方法程序是 SetAll 方法，用于为按钮组中的所有控件或某类控件指定

一个属性。该方法的格式是：容器.SetAll(cProperty,Value [,cClass])，这在第 5 章已经作了介绍。例如，

要将命令按钮组 1 中所有的命令按钮设置为不可用的命令为： 
Thisform.CommandGroup1.SetAll("Enabled",.F.,"CommandButton")   

命令按钮组的常用事件与命令按钮一样，主要是单击（Click）事件。如果是用命令按钮组，那么

对于按钮组里每个按钮的Click事件代码一般让组的Click事件代码来统一管理。用命令按钮组的Value
（指明用户按了哪个按钮）属性和 DO CASE 结构来实现，如下面的代码所示： 

(a)                                                    (b) 

图 6-18  命令按钮组的生成器 

DO CASE 
CASE THIS.Value=1 
    Command1的 Click事件代码  
CASE THIS.Value=2 
    Command2的 Click事件代码 
CASE THIS.Value=3 
    Command3的 Click事件代码 
… 
ENDCASE 

应当注意的是，如果命令按钮组和其中命令按钮的 Click 事件中都含有代码，则当单击该按钮时

触发的是按钮的 Click 事件代码，而不是命令按钮组的 Click 事件代码。只有当按钮的 Click 事件中不

包含代码时才触发按钮组的 Click 事件代码。 
【例 6.9】编写一个无纸化考试的表单，运行界面如图 6-19 所示。题目内容取之于表文件的备注字

段，学生答题时能将所选答案存入表文件对应字段中。表文件里有标准和得分字段，可以统计学生的

总分。 
（1）问题分析 
① 根据题目的要求，先建立一个表文件，用来存放题目、答案等数据，把这张表命名为“题目

及答案”，表结构如图 6-20 所示。因为题目文字较多，且长短不一，所以设置“题目”字段为备注型，

“答案”字段用于存放用户选择的结果，“标准”字段存放标准答案，用于要编制机器改卷程序时备用。 

              
           图 6-19  例 6.9 的运行界面                  图 6-20 “题目及答案”表文件的结构 

② 表单设计界面如图 6-21 所示，其中使用了 1 个编辑框、1 个标签、1 个选项按钮组、1 个命令

按钮组。选项按钮组将在 6.3.2 节介绍，本例中先不考虑选项按钮组，重点设计命令按钮组及其 Click
事件代码。用编辑框只读显示备注字段的内容，设置编辑框的ControlSource为“题目”字段，其ReadOnly
为.T.即可。 
（2）设计步骤 
① 新建表文件“题目及答案”，按照图 6-20 所示的要求输入表结构，“题目”和“标准”字段的

数据自编或从电子数据复制过来。 
② 新建表单，“题目及答案”表文件放入数据环境。放入控件：1 个编辑框、1 个标签、1 个选项

按钮组、1 个命令按钮组。 

③ 在“属性”窗口设置属性：设置表单和标签的 Caption，设置标签的 FontName、 FontSize 为

适当，设置编辑框的 ControlSource 为“题目及答案.题目”字段，设置编辑框 ReadOnly 为.T.，调整编

辑框的字体为适当大小。 
④ 用命令按钮组的生成器设置其按钮数为 5，在生成器对话框中设置每个按钮的标题并给每个按

钮加上快捷键（标题中\<字母，按 Alt+字母可以操作该按钮），如图 6-22 所示，再设置按钮布局为水

平，按钮间距为 10。用命令按钮组快捷菜单的“编辑”命令进入各按钮的编辑状态，选中全部的按钮

设置它们的字体为适当大小。 

          
                  图 6-21  例 6.9 的设计界面            图 6-22  用命令按钮组生成器设置按钮 

⑤ 编写命令按钮组 CommandGroup1 的 Click 事件代码如下： 
DO CASE  
 CASE This.Value=1       &&单击"首记录"按钮 
  GO TOP               &&指针定位到首记录 
  This.Command3.Enabled=.T.  &&"下一题"按钮能使用 
  This.Command4.Enabled=.T.  &&"末记录"按钮能使用 
  This.command1.Enabled=.F.  &&"首记录"按钮不能使用 
  This.Command2.Enabled=.F.  &&"上一题"按钮不能使用 
 CASE This.Value=2.AND.!BOF()   &&单击"上一题"按钮 
  SKIP -1               &&指针上移一条记录 
  This.Command3.Enabled=.T.  &&"后翻"按钮能使用 
  This.Command4.Enabled=.T.  &&"末记录"按钮能使用 
  IF BOF()          &&如果到文件头 
   This.Command1.Enabled=.F.   &&"首记录"按钮不能使用 
   This.Command2.Enabled=.F.   &&"上一题"按钮不能使用 
  ENDIF 
 CASE This.Value=3.AND.!EOF()    &&单击"下一题"按钮 
  SKIP                  &&指针下移一条记录 
  This.Command1.Enabled=.T.  &&"首记录"按钮能使用 
  This.Command2.Enabled=.T.  &&"上一题"按钮能使用 
  IF EOF()       &&如果到文件尾 
   This.Command3.Enabled=.F.   &&"下一题"按钮不能使用 
   This.Command4.Enabled=.F.   &&"末记录"按钮不能使用 
  ENDIF 
 CASE This.Value=4       &&单击"末记录"按钮 
  GO BOTTOM            &&指针定位到末记录 
  This.Command1.Enabled=.T.  &&"首记录"按钮能使用 
  This.Command2.Enabled=.T.  &&"上一题"按钮能使用 
  This.Command3.Enabled=.F.  &&"下一题"按钮不能使用 
  This.Command4.Enabled=.F.  &&"末记录"按钮不能使用 
 CASE This.Value=5         &&单击"退出"按钮 
  Thisform.Release      &&释放表单   
ENDCASE 
Thisform.Refresh 

6.3.2  复选框与选项按钮组 

1．复选框 
复选框（CheckBox）在控件工具栏中的图标是 。按照复选框放入表单的顺序，复选框的默认

名称为 Check1、Check2 等。 
复选框的功能是用于指定一个值的两种状态：“真”、“假”；“开”、“关”；“是”、“否”等。将多

个复选框组合在一起可以对一个问题进行多项选择。有时不能将问题准确地归为“真”或“假”，复选

框还有一个中间状态。 
复选框的三种状态是：选中状态（Value 值为 1 或.T.）、未选中状态（Value 值为 0 或.F.）、中间状

态（Value 值为 2 或.NULL.）。复选框的 Value 属性可以设置为数值型，可用值为 0、1、2；也可以设

置为逻辑型，可用值为.F.、.T.、.NULL.。三种状态的外观如图 6-23 所示。 
通常情况下，在运行时刻只能通过鼠标单击来选择复选框为 0（.F.）或 1（.T.），如果希望复选框

的 Value 值为 2，可以按 Ctrl+0 组合键。 
复选框的常用属性如表 6-12 所示，其中 Themes 用于指定复选框的外观效果。 

表 6-12  复选框的常用属性 

属性名称 功能说明 取值说明 

Value 指定复选框的当前值 0、1、2 或.F.、.T.、.NULL.，默认为 0 
ControlSource 指定复选框的数据源 表的逻辑型或数值型字段、内存变量 
Themes 指定复选框是否为 Windows XP 主题效果 .T.（默认），.F. 

【例 6.10】设计一个统计车票总金额的表单，要求是：在给出的各种车票中选择所需要的，按

“计算总额”命令按钮，将所需要的车票总金额计算出来并显示在指定的文本框中。设计界面如图 6-24
所示。 

            
图 6-23  复选框的三种状态        图 6-24  例 6.10 设计界面 

（1）问题分析 
多种车票用多个复选框列出便于用户灵活选择。注意，复选框 Value 的默认初值为 0，是数值型

的。只有当设置了复选框的 ControlSource 属性为逻辑字段或设置了 Value 的初值为逻辑型的数据后，

Value 的数据类型才可以变为逻辑型。 
如果复选框选中，即 Value 值为 1（不是为.T.），则将对应的票价加起来。对于每个复选框都需要

判断，最后把相加的结果显示在 Text1 中。以上的处理写在“计算总额”命令按钮的 Click 事件中。 
（2）设计步骤 
① 新建表单，放入 2 个标签、3 个复选框、2 个命令按钮、1 个文本框。 
② 在“属性”窗口设置属性：2 个标签、3 个复选框、2 个命令按钮及表单的 Caption 属性。设置

“计算总额”命令按钮为默认按钮，即 Default 为.T.。设置文本框 Text1 的只读属性 ReadOnly 为.T.，选

中所有对象，设置字体为适当的大小。 

③ 编写命令按钮 Command1（计算总额）的 Click 事件代码如下： 
S=0       &&为放总额的变量赋初值 0 
IF Thisform.Check1.Value=1 &&如果 Check1被选中 
 S=S+580 
ENDIF 
IF Thisform.Check2.Value=1 &&如果 Check2被选中 
 S=S+50 
ENDIF 
IF Thisform.Check3.Value=1 &&如果 Check3被选中 
 S=S+280 
ENDIF 
Thisform.Text1.Value=S  &&显示总金额在 Text1中 
Thisform.Refresh 

④ 命令按钮 Command2（退出）的 Click 事件代码如下： 
Thisform.Release 

2．选项按钮组 
选项按钮组（OptionGroup）在控件工具栏中的图标是 。按照选项按钮组放入表单的顺序，复

选项按钮组的默认名称为 OptionGroup1、OptionGroup2 等。 
选项按钮组的功能主要是允许用户在给定的多个选项中选中一个，且只能选一个。与复选框控件

不同，复选框控件独立存在于表单中，可以一次选择多项。选项按钮组是一个容器类对象，它里面包

含多个选项按钮，一次只允许选择一项。 
选项按钮组的常用属性如表 6-13 所示。 

表 6-13  选项按钮组的常用属性 

属性名称 功能说明 取值说明 

ButtonCount 指明按钮组中选项按钮的数目 正整数，默认为 2 
Buttons(i)  用于引用每个按钮的数组  i:1～ButtonCount 
Value 指明用户按了哪个按钮  正整数：1～ButtonCount 
ControlSource 指定选项按钮组的数据源  表的字段、内存变量 

其中 Value 属性的值，对应的是在选项按钮组中选中的按钮。例如，当 Value 值为 1，表示在选项

按钮组中选中了第 1 个按钮等。 
在设计选项按钮组时，可以利用 Visual FoxPro 提供的选项按钮组生成器，方便地设置按钮的个数、

按钮的标题及按钮的布局等。这与命令按钮组生成器的使用类似。 
选项按钮组的常用事件主要是 InterActiveChange 事件，该事件在用户改变了选项按钮组的值时触发。 
【例 6.11】利用选项按钮组设计调色板表单，能调出红、橙、黄、绿、青、蓝、紫、黑、白 9 种

颜色。初始运行颜色为白色。设计界面和运行界面分别如图 6-25 和图 6-26 所示。 

    
图 6-25  例 6.11 设计界面    图 6-26  例 6.11 运行界面 

（1）问题分析 
本题选项按钮组里包含 9 个按钮，按三行三列排列。颜色是用选项按钮组的背景色调出来的。改


